Move =(+0,+5): Making SAS/GRAPH® Work For You

Deb Cassidy, Computer Horizons Corporation, Indianapolis, IN

I've often been asked "Can SAS/GRAPH® do...?" SAS/GRAPH can do almost anything - if you are willing to take the time. This paper shows you some of the options for enhancing your graph.

The following code created a basic graph in BASE SAS®.

PROC PLOT DATA=MWSUG;
   PLOT ATTEND*CONF_NO;
TITLE1 'SAS USERS ATTENDING MWSUG';
FOOTNOTE 'NOT ACTUAL DATA'
     'MWSUG SEPT93';

This graph probably would not convince your boss to send you to the MidWest conference.

GOPTIONS

SAS/GRAPH lets you change many options. Before you begin a graph, make sure you know how options are set for your device. You will need to be concerned with both the viewing device and printing device. Options are specified with a GOPTIONS statement. The one option everyone must specify is DEVICE=. SAS prompts you if you don't tell it the device. You may not need any other options or you may need several. Some sites have created macros to supply all the options for a device so individuals do not need to enter them. You can override options with another GOPTIONS statement. You can also override some within the statements for your graph. To know what options you device has as defaults, run

GOPTIONS DEVICE= your device;
PROC GOPTIONS;

If you want to know what statements are in effect at a specific point in your program, you should specify the statement(s) you want to see. For example:

PROC GOPTIONS SYMBOL AXIS NOLIST;

will give you the SYMBOL and AXIS statements while suppressing the list of graphics options.

Options stay in effect until you reset them or end your SAS session (for batch and non-interactive, each job would be a session). Any options listed before a RUN or a PROC will affect the graph. RUN or PROC tells SAS to go ahead and created the graph. Many times you won't get what you expected because of an earlier option that is still in effect or because you didn't set an option that you had in a previous session. I suggest always putting options before a graph and using RUN after each graph. However, too many RUN statements can cause problems. If you get multiple copies of a graph, look for duplicate RUN statements. This is most likely to happen if you have a RUN within a macro. If you rely on PROC statements to signal the end of a graph, you may need a RUN statement after the last graph.

ENHANCED GRAPH

The following code creates the enhanced graph shown. The first step should be to run the graph without any enhancements to make sure you have the basic graph correct. Then you should work through each enhancement to verify how it changes the graph. As you become more familiar with the options, you can put more in each run. However,
sometimes you will not be able to use some options with others. For example, if your titles and footnotes are too tall, you will not get your graph. You may need to change all the heights and widths I've shown if you have a different HPOS and VPOS for your device.

GOPTIONS DEVICE=your device
   HPOS=80 VPOS=32
   COLORS=(BLACK) PTEXT=SWISS;

PROC GPLOT DATA=MWSUG;
   PLOT ATTEND*CONF_NO /
     VAXIS=AXIS1 HAXIS=AXIS2
     VMINOR=1 HMINOR=0;

AXIS1 ORDER=(200 TO 550 BY 50)
   LABEL (A=90 'ATTENDANCE')
   WIDTH=2 OFFSET=(2 PCT, 2.2 PCT);

AXIS2 VALUE=
   (TICK=1 J=C 'MINNEAPOLIS'
    J=C '1990'
   J=C 'CHICAGO'
   J=C '1991',
   TICK=3 J=C 'KANSAS CITY'
   J=C '1992',
   TICK=4 H=1.5
   J=C 'INDIANAPOLIS'
   J=C '1993');

LABEL=(H=1.5 'CONFERENCE')
   WIDTH=2 OFFSET=(3 PCT);

SYMBOL1 H=3 FONT=, VALUE=DIAMOND
   I=JOIN WIDTH=2;

TITLE1 'SAS' MOVE=(+0, +1 PCT)
   F=SPECIAL 'R' MOVE=(+0, -1 PCT)
   F=CENTBI 'USERS ATTENDING'
   F=CENTBI 'MWSUG';

TITLE2 H=3.3 A=-90 ' ';

NOTE MOVE=(71 PCT, 50 PCT)
   'FUTURE CONFERENCES'
   DRAW=(+0 PCT, -1 PCT)
   MOVE=(71 PCT, 46 PCT)
   '1994 - OMAHA'
   MOVE=(71 PCT, 43 PCT)
   '1995 - SOME CITY');

NOTE MOVE=(84, 78.5 PCT)
   BOX=2 H=.5 F=DUPLEX 'EXPECTED';

FOOTNOTE H=.5 J=L 'FICTITIOUS DATA'
   J=R 'MWSUG -- SEPT 93';
RUN;

The plot shows the attendance on the Y-axis (vertical) and the conference number on the X-axis (horizontal).

VAXIS1=AXIS1 says to create the vertical axis using the instructions in the AXIS1 statement. HAXIS=AXIS2 says to create the horizontal axis using the instructions in the AXIS2 statement. You do not have to use a specific axis number for vertical or horizontal. Some graphs have 2 horizontal or vertical axes. AXIS numbers can be 1 to 99 although most people start with 1.

AXIS1 says to start at 200 and go to 550 by 50 unit increments. In the example, one value if at 505. If you try to use ORDER=(200 TO 500 BY 50), you will lose that point. You could also use ORDER=(200 TO 500 BY 50, 505) which says to put a tick mark at 505 also. However, the distance between 500 and 505 will be the same as between 450 and 500.

LABEL=(A=90 'ATTENDANCE') says to use ATTENDANCE instead of the variable name. A=90 angles the entire label from the horizontal line. You may need to use A=90 and R=0 together. R= says to ROTATE each letter. In this case, you are over-riding any default rotation. I've usually had this problem if there is already a label for the variable in my data set.

WIDTH=2 makes the axis itself wider. If you have a data point in the corner of the graph, you may want to use OFFSET= to give it more room. The first value moves the first major tick mark away from the intersection of the axes. The second value moves the last major tick mark away from the end of the axis. You can use one without the other. OFFSET only affects the axis specified.
The AXIS2 statement uses some of the same options as AXIS1. I've used the default order although I could have specified (1 TO 4) or (1 TO 4 BY 1). I've also labeled the major tick marks. You can use a format to label the tick marks but you have fewer options. I wanted to create a 2-line label. I also wanted to increase the height for the current site. This method lets me customize each tick mark. To create multiple lines, you must center each section of the label. This also works for X-axis labels. SAS Institute is aware that it doesn't always work for Y-axis labels.

The plot statement has two other options - VMINOR=1 and HMINOR=0. These statements indicate how many minor tick marks to include. On the vertical axis I want one smaller tick mark between each major one. In this example, that provides marks every 25 units. Since my conference numbers are discrete, I want to suppress minor tick marks on the horizontal axis. You could also have used MINOR=NONE in the AXIS2 statement to delete them. You also have the options of customizing minor tick marks.

SYMBOL describes how to put symbols on the points. By default, you get a symbol on each point and the points are not joined. I want the symbols to have HEIGHT=3. I want to INTERPOLATE the points between the actual points by simply JOINing the points. There are many other options for INTERPOLATE. The line joining the points will have a WIDTH of 2. I also want to specify the symbol. There are special symbols that are only used in a symbol statement. You can also use any character by specifying a font and placing the letter in quotes. For example:

\texttt{FONT=SPECIAL \textsc{VALUE}='R'}

would give the registered trademark symbol on the graph. If you don't specify the font, SAS looks for the specified value as one of the special symbols. If it isn't used as a special symbol, you get the actual value specified. By using \texttt{VALUE=DIAMOND}, I get a diamond as the symbol. If there is a font specified and I use \texttt{VALUE=DIAMOND}, I actually get the word DIAMOND. Symbols and patterns will keep any option you used in an earlier statement with the same number. If I had an earlier SYMBOL1 \texttt{FONT=DUPLEX} statement, I would get the word DIAMOND in duplex. The \texttt{FONT=}, included in this symbol statement makes sure I get the special symbol. If this sounds confusing, use \texttt{PROC GOPTIONS SYMBOL NOLIST;} to see what SAS "sees".

The first title statement shows how I named this presentation. I wanted the registered trademark symbol to be superscript. MOVE= tells SAS to go to a different point before writing the next section of text. A + or - indicates this point is relative to where the last text was written. No sign indicates to go to that point regardless of the location of the current text. The first value indicates the horizontal direction and the second value the vertical direction. I wanted to slide the character up but keep it in the default horizontal direction. The +0 indicates to keep the default. I wanted to move the text up 1 percent. After I finish with the character, I wanted to move the next text back to the original line so I use -1 percent. Don't forget to include the blank spaces in one of the strings. I also used different fonts. If you want to move the first string and use a relative position, you must use \texttt{TITLE \textsc{MOVE}=...;} to get a starting point.
TITLE2 is a blank title so why did I include it? By using A=-90, I moved the title to the right-hand side of the graph. This makes the graph smaller and gives me a larger right margin. The NOTE statement will then appear in that margin without overwriting any of the graph. The HEIGHT= determines how much additional space I have. I could have used OFFSET the move the plot itself over but the axis would still be there.

NOTE statements may be one of the most useful and least used statements in SAS/GRAPH. In some cases, NOTES can let you avoid creating an ANNOTATE data set. By default, NOTES are written in the procedure output area. They are left-justified and placed at the top of the area. They will overwrite your graph instead of reducing the size of the graph like TITLE and FOOTNOTE do. You can have any number of NOTES and they do not have a number. I rarely use NOTE without a MOVE= to position it exactly where I want.

I have only 1 note statement but it prints 3 lines of text and draws a line. The MOVE= works like it did in the TITLE statement but it uses absolute values. The tricky part is drawing the line. The line is actually drawn from right to left. To draw a line, you need to specify the ending point. You know the left-hand point is just below the starting point of your text - a point you specified. However, you don't know the ending point of your text. By starting relative to the end of the text, you only need to specify the X value for the left end of the line.

If you want to box the entire note, you can only use relative points in your MOVE. Finding the correct coordinates can be frustrating if you have more than a couple lines. I recommend using an ANNOTATE data set for that case.

The second note does use a BOX= option. The value specifies the width of the lines for the box. This note actually appears in the graph area. I've also reduced the HEIGHT of the text and changed the font.

The footnote statement left justifies part of the text and right justifies the rest. The default is centered text.

A HELPFUL GRID

You think MOVE= will be helpful but how do you know what values to use without a lot of guesswork? The following code will generate a grid which can be placed over your graph to help determine coordinates. The grid is based on percentage of the graphics output area. You might also want a grid which shows your cells based on HPOS and VPOS. You will need to change the DO loop accordingly. For each line you want to draw, your annotate data set needs one observation that moves to one end of the line. A second observation is needed that will draw to the other end of the line. The grid is displayed using PROC GANNO.

DATA ANNO;
/* X AND Y VALUES BASED ON
   PERCENTAGE OF THE GRAPHICS
   OUTPUT AREA */
RETAIN XSYS '3' YSYS '3';
DO I=0 TO 100 BY 10;
/* DRAW VERTICAL LINES */
X=I; Y=0; FUNCTION='MOVE'; OUTPUT;
Y=100; FUNCTION='DRAW'; OUTPUT;
/* DRAW HORIZONTAL LINES */
Y=I; X=0; FUNCTION='MOVE'; OUTPUT;
X=100; FUNCTION='DRAW'; OUTPUT;
PROC GANNO ANNO=ANNO
GOUT=MWSUG.PLOTS NAME=GRID1;
RUN;

806
OTHER COMMENTS

The values needed for height and width will vary based on HPOS and VPOS. PERCENT is the only value that is consistent across devices and options because it is proportional. Even inches and centimeters may not work if you go to a smaller device or changes options.

Since it is impossible to cover all options, you should use the manuals for additional details. The programs for examples in the manual are included in a library provided with the system. Contact your site representative for details on how to access this library. The manual indicates a program name. These programs can help you learn how your device(s) work (and differ) as well as how to write the program.

At times, using SAS/GRAPH is trial and error. But remember, what didn't work this time may be the solution to your next graph. You may want to keep a notebook for both your successes and failures. Include the code, log and the graph.

SAS and SAS/GRAPH are registered trademarks of SAS Institute Inc. in the USA and other countries.

Contact:
Deb Cassidy
3809 Secretariat Court
Columbus, OH 43221
(614) 771-7861
(leave message on answering machine)

or
Computer Horizons Corporation
1980 E. 116th Street, Suite 325
Carmel, IN 46032
(317) 574-5600

This paper is part of a presentation given at the 1993 MidWest SAS Users Group conference. The proceedings contain additional examples of enhancing graphs.

DATA FOR THE GRAPH

DATA MWSUG;
  INPUT CONF_NO ATTEND;
CARDS;
  1 200
  2 500
  3 450
  4 505
;
SAS® USERS ATTENDING MWSUG

Plot of ATTEND*CONF_NO. Legend: A = 1 obs., B = 2 obs., etc.

FICTITIOUS DATA  MWSUG -- SEPT 93

SAS® USERS ATTENDING MWSUG

EXPECTED

FUTURE CONFERENCES
1994 -- OMAHA
1995 -- SOME CITY

MINNEAPOLIS 1990
CHICAGO 1991
KANSAS CITY 1992
INDIANAPOLIS 1993

CONFERENCE

FICTITIOUS DATA  MWSUG -- SEPT 93